What?

Management of our financial portfolio needs to be evaluated. Have we been successful or have we failed?

Obviously, having positive returns usually translates as having success. However, as part of our portfolio management, we are not only interested in positive returns. It is important to assess whether we followed our investment policies and whether we reached our investment goals.

Just as important is our relative success or failure. Our risk and return will be compared to those of major benchmarks.

Our portfolio

Let's start with defining our portfolio. We will include an equity ETF (DIA) with a portfolio weight of 70% and a bond ETF (BND) with a weight of 30%.

Jensen's Alpha

One of our performance ratios is the Jensen's alpha. Jensen's alpha is based on the undiversifiable market risk (i.e. Beta). With this ratios, we evaluate how much return we realize above and beyond what was expected based on our portfolio's correlation with the overall market.

Let's evaluate our portfolio's performance with respect to Jensen's alpha. We will compare our performance to several popular equity ETFs. SPY is the SPDR S&P 500 ETF. IJH is the iShares Core S&P Mid-Cap ETF. IJR is the iShares Core S&P Small-Cap ETF. OEF is the iShares S&P 100 ETF. QQQ is the Invesco QQQ Trust (NASDAQ). IWM is the iShares Russell 2000 ETF. IYR is the iShares U.S. Real Estate ETF. VEU is the Vanguard FTSE All World Ex US ETF.

```
local symbols = "BND DIA SPY IJH IJR OEF QQQ IWM IYR VEU"
fetchyahooquotes `symbols', freq(d) chg(ln) start(31dec2017)
gen ln_portfolio = (ln_BND*0.30)+(ln_DIA*0.70)
collapse (sum) ln*
save temp1.dta, replace
fetchyahooquotes `symbols', freq(d) chg(ln) start(31dec2017)
gen ln_portfolio = (ln_BND*0.30)+(ln_DIA*0.70)
foreach aa in `symbols' "portfolio" {
        reg ln_`aa' ln_SPY
        mat temp=r(table)
        local beta_`aa' = temp[1,1]
        mat drop temp
}
collapse (sd) ln*
append using temp1.dta
erase temp1.dta
xpose, clear varname
rename v1 sd
rename v2 return
rename _varname symbol
replace symbol = subinstr(symbol, "ln_", "", .)
order symbol sd return
gen beta=.
foreach aa in `symbols' "portfolio" {
        replace beta = `beta_`aa''  if symbol=="`aa'"
}
gen market_return = return if symbol=="SPY"
summ market_return
replace market_return = r(mean)
gen exp_return = market_return * beta
gen jensen_alpha = return - exp_return
gsort -jensen_alpha
browse symbol jensen_alpha
```

symbol	jensen_alpha
IYR	.0278391
QQQ	.021361
DIA	.0110872
IJR	.0061183
OEF	.0032826
portfolio	.0027154
SPY	0
BND	-.0168187
IWM	-.030281
IJH	-.037384
VEU	-.1452701

Notice that, with respect to the Jensen's alpha, our portfolio outperformed five of the equity ETFs. These results are specific to 2018.

Let's test our portfolio during 2017. This would be referred to as out-of-sample testing.

```
local symbols = "BND DIA SPY IJH IJR OEF QQQ IWM IYR VEU"
fetchyahooquotes `symbols', freq(d) chg(ln) start(31dec2016) end(31dec2017)
gen ln_portfolio = (ln_BND*0.30)+(ln_DIA*0.70)
collapse (sum) ln*
save temp1.dta, replace
fetchyahooquotes `symbols', freq(d) chg(ln) start(31dec2016) end(31dec2017)
gen ln_portfolio = (ln_BND*0.30)+(ln_DIA*0.70)
foreach aa in `symbols' "portfolio" {
        reg ln_`aa' ln_SPY
        mat temp=r(table)
        local beta_`aa' = temp[1,1]
        mat drop temp
}
collapse (sd) ln*
append using temp1.dta
erase temp1.dta
xpose, clear varname
rename v1 sd
rename v2 return
rename _varname symbol
replace symbol = subinstr(symbol, "ln_", "", .)
order symbol sd return
gen beta=.
foreach aa in `symbols' "portfolio" {
        replace beta = `beta_`aa''  if symbol=="`aa'"
}
gen market_return = return if symbol=="SPY"
summ market_return
replace market_return = r(mean)
gen exp_return = market_return * beta
gen jensen_alpha = return - exp_return
gsort -jensen_alpha
browse symbol jensen_alpha
```

symbol	jensen_alpha
DIA	.08167
portfolio	.0751995
VEU	.0724376
BND	.0601016
QQQ	.0454652
OEF	.0124673
SPY	0
IYR	-.0216719
IJH	-.0988872
IWM	-.1469424
IJR	-.1607119

For the year 2017, our portfolio outperforms almost all our benchmarks.